
From: peterh5322@... (Present e-mail: peterhaas@cruzio.com) 11/20/2008 PHH
Date: Sun Jan 1, 2006 12:03 pm 
Subject: Re: [hercules-390] Re: looking for info on excp+ccw
 how to write, startio
 peterh5322 
 Send Email  
  
In a message dated 12/31/05 4:44:19 PM, somitcw@... writes:

> A friend of mine borrowed an amdahl
> Tech.Topic white paper from me and keyed
> the sample up. He didn't know assembler so
> may have a keying error. The code is at:
>
> http://groups.yahoo.com/group/H390-MVS/files/startio.txt

I am the author, and the owner of this Intellectual Property (IP),
a facsimile of which (see 'Net reference cited above) has been
modified to delete the author and ownership notice, and, indeed,
that it was, in fact, an Amdahl Technical Topic. Issue 2, to be
specific.

I was the originator of the Amdahl Technical Topics series.

I was its editor, and I was the sole author of this issue
(there were other authors of other Technical Topics issues).

What has been posted to Hercules-390@yahoogroups.com is not a
precise copy of my IP, but is essentially a 99.44 percent copy,
with the author and ownership notice deleted, as was mentioned.
There appear to be a few other alterations of a non-substantive
nature.

This Technical Topic was originally written by me in about 1975,
and was first published under the Amdahl Technical Topics
imprimatur in about 1977.

Amdahl Technical Topics were disclosed to Amdahl Corporation's
Product Support and Services group employees (PS&S), primarily
through Amdahl's MVS Support Team (MVSST; functionally a part
of PS&S), of which I was a Headquarters member, although any
Amdahl system engineer could order this publication through
Amdahl's Technical Publications group (Tech Pubs).

Subsequent disclosure to Amdahl's customers was encouraged,
where there was a compelling business reason to do so.



For reasons which remain unclear to me to this day, every
Amdahl Technical Topic except the instant one, was copyrighted
by Amdahl Corporation. The instant one, "The STARTIO Facility
of MVS", was then, and remains copyrighted by me.
This is possibly because a significant portion of this IP,
if not the IP in its entirety excepting certain text of a
non-substantive nature, was completed before I began my
employment at Amdahl Corporation. These facts were known by
Amdahl Corporation's officials.

After the MVSST was disbanded, roughly in 1980, publication
of Amdahl Technical Topics ceased, although the then existing
stock within Tech Pubs remained available as described, until
such stock was exhausted.

No reprints were made, to my knowledge, and the pre-print
masters were not archived, although I formally requested
that the masters of my Technical Topics be returned to me.

I believe the masters of all Amdahl Technical Topics were
destroyed as an economy measure.

If anyone has an original issue of "The STARTIO Facility of
MVS", I would surely like to have it for my personal archives.

You know how to contact me.

Once so archived, I will be pleased to formally release the
original it to anyone who wishes to use it.

Peter.

//SOMITCWS JOB (xxxxxxxx,xxxx,1439,9999,9999),STARTIO-SOMITCW,MSGCLASS=x,
// CLASS=x,NOTIFY=SOMITCW
//*
//* Warning; This was keyed up from an amdahl Tech. Topic
//* It could have typos.
//*
//  EXEC ASMFCL,MAC1='SYS1.AMODGEN'    ,PARM.LKED='AC=1'
//SYSPRINT DD  SYSOUT=*
STARTIO  CSECT ,
         PRINT NOGEN
*



*
*    PROGRAM NAME--
*    STARTIO
*
*    PURPOSE--
         SPACE 3                            PSA ADDRESSABILITY
         USING PSA,0                        SAVE REGISTERS
         BALR  12,0
         USING *,12
         SPACE 2
*
*
*
*    PROGRAM INITIALIZATION.  A DYNAMICALLY ACQUIRED WORKAREA IS
*    IS OBTAINED ON A PAGE BOUNDARY (THIS SIMPLIFIES BUILDING THE REAL
*    CHANNEL PROGRAM.)  THE MODESET MACRO IS ISSUED TO CHANGE TO
*    THE SRB IS INITIALIZED TO CONTAIN A POINTER TO THE CURRENT TCB
*    AND THE ADDRESS OF THE IOSB.  (THE TCB-0 OPTION OF THE STARTIO
*    MACRO COULD ALSO BE USED.  REFER TO STARTIO MACRO COMMENTS FOR
*    DETAILS.
*
*
         SPACE 2
         GETMAIN RU,LV=CORESIZE,BNDRY=PAGE
         XC    0(CORESIZE,1),0(1)
         ST    13,4(1)                  CHAIN SAVE AREA
         ST    1,8(13)
         LR    13,1
         USING CORE,13                  DYNAMIC STRORAGE ADDRESSABILITY
*   OBTAIN CVT ADDRESS
         L     3,FLCCVT
         USING CVT,3                    CVT ADDRESSABILITY
*   OBTAIN DYNAMIC STORAGE FOR SRB/IOSB
         GETMAIN RU,LV=SRBSIZE+IOSBSIZE,SP=245
         XC    0(SRBSIZE+IOSBSIZE,1),0(1)
         LR    4,1
         USING SRB,4                   SRB ADDRESSABILITY
         LA    5,SRB+SRBSIZE
         USING IOSB,5
         L     6,PSAAOLD
         USING ASCB,6
*   INITIALIZE SRB
         MVC   SRBPTCB,PSATOLD
         ST    5,SRBPARM
         SPACE 2
*



*
*   INITIALIZE THE IOSB.  MOST OF THE FIELDS INITIALIZED ARE SELF-
*   EXPLANATORY.  AN EXCEPTION IS THE FIELD IOSDVRID.  IT IS USED
*   AS AN INDEX INTO THE VECTOR OF IOS DRIVERS (VOID) TABLE.  THIS
*   IS WHERE IOS' DIRECT ACCESS ERROR RECOVERY COMPONENT FINDS THE
*   ADDRESS OF THE EXTENT CHECKING SUBROUTIN.  A RETURN +0 FROM
*   THIS SUBROUTINE INDICATES THE UPDATED SEEK ADDRESS VIOLATES
*   EXTENTS.  THIS IS ALSO WHERE IOS' PURGE AND RESTORE COMPONENTS
*   FIND THE ADDRESSES OF THE PURGE AND RESTORE SUBROUTINES.  IN THIS
*   EXAMPLE,  THE MISCELLANEOUS DRIVER ID IS INDICATED.   IT'S VOID
*   ENTRY CONTAIN THE ADDRESSES OF DUMMY (NO-OP) SUBROUTINES.
*
*
         SPACE 2
         MVI   IOSFLA,IOSCCHN          COMMAND CHAINING
         MVI   IOSDVRID,IOSMISID       DRIVER ID
         MVC   IOSASID,ASCBASID
         LA    1,STIOPSTR              TERMINATION ROUTINE ADDR.
         ST    1,IOSPGAD
         MVC   IOSUCB,CVTSYSAD         SYSRES UCB ADDR.
         LA    1,CORE                  DYNAMIC CORE ADDRESS
         ST    1,IOSUSE                PARMADDR. FOR TERMINATION
         LA    1,STIONRMR              ADDR. OF NORMAL EXIT
         ST    1,IOSNRM
         LA    1,STIOABNR              ADDR. OF ABNORMAL EXIT
         ST    1,IOSABN
         MVI   IOSFMSK,X'40'           FILE MASK
         MVI   IOSCKEY,X'80'           PROTECTION KEY
         MVC   IOSEEK,=X'0000000000000003'  SEEK ADDR.
         MVC   IOSEEKA,=X'00000000000000003'  SEEK ADDR.
         SPACE 2
*
*
*    FIX THE CHANNEL PROGRAM AND I/O AREA.  THE PGFIX MACRO IS USED.
*    IN HIGH PERFORMANCE APPLICATIONS THE PGFIX BRANCH ENTRY IS
*    RECOMMENDED.
*
*
         SPACE 2
         LA    0,ECB                   ECB ADDR.
         LA    1,FIX                   STARTING LOCATION OF FIX
         LA    15,FIXEND               ENDING LOCATION + 1
         XC    ECB,ECB
         PGFIX R,ECB=(0),A=(1),EA=(15),LONG=N
         WAIT  ECB=ECB                 WAIT FOR COMPLETION
         SPACE 2



*
*
*    OBTAIN THE LOCAL LOCK.  PRIOR TO INVOKING BASIC IOS, A REAL CHAN-
*    NEL PROGRAM IS BUILT IN DYNAMICALLY ACQUIRED CORE.  THE REAL AD-
*    DRESSES OBTAINED BY THE LRA INSTRUCTION MUST NOT CHANGE UNTIL THE
*    I/O IS INITIATED BY BASIC IOS.  THE LOCAL LOCK PREVENTS ANY
*    ACTIVITY IN THE MEMORY WHICH COULD RESULT IN THIS CHANGE.
*
*
         SPACE 2
         SETLOCK OBTAIN,TYPE=LOCAL,    OBTAIN LOCAL LOCK               X
               MODE=UNCOND,                                            X
               REGS=USE,                                               X
               RELATED=(STARTIO,RELEASE)
         SPACE 2
*
*
*    BUILD THE REAL CHANNEL PROGRAM.  THE CHANNEL PROGRAM CONSISTS OF
*    A SEARCH ID EQUAL - TIC - READ SEQUENCE.  THE SEARCH ARGUMENT IS
*    CONTAINED IN THE IOSB AT LOCATION IOSEEK+3.  ALL OTHER DATA IS
*    LOCATED IN DYNAMICALLY ACQUIRED STORAGE.
*
*
         SPACE 2
         LRA   1,IOSEEK+3
         ST    1,SEARCH
         MVI   SEARCH,X'31'
         MVC   SEARCH+4(4),=X'40000005'
         LRA   1,SEARCH
         ST    1,TIC
         MVI   TIC,X'08'
         MVC   TIC+4(4),=X'00000000'
         LRA   1,LABEL
         ST    1,READ
         MVI   READ,X'06'
         MVC   READ+4(4),=X'00000050'
         SPACE 2
*
*
*    INSERT ADDRESS OF CHANNEL PROGRAM INTO IOSB.  THE REAL ADDRESS
*    IS LATER MOVED INTO THE CHANNEL ADDRESS WORD.  THE VIRTUAL AD-
*    DRESS IS FOR USE BY EXIT ROUTINES.  UPON COMPLETION OF THE I/O,
*    BASIC IOS TRANSLATES THE CHANNEL STATUS WORD (WHICH CONTAINS A
*    REAL ADDRESS) INTO A VIRTUAL ADDRESS BY INVOKING THE RESIDENT
*    CONVERT REAL TO VIRTUAL ADDRESS ROUTINE.  THIS ROUTINE IS AC-
*    CESSED VIA CVTPRTV.



*
*
         SPACE 2
         LRA   1,SEARCH                CHANNEL PROGRAM REAL ADDRESS
         ST    1,IOSRST
         LA    1,SEARCH                CHANNEL PROGRAM VIRTUAL ADDR.
         ST    1,IOSVST
         SPACE 2
*
*
*    ISSUE STARTIO MACRO TO INVOKE CHANNEL SCHEDULER.  BASIC IOS (ALSO
*    KNOWN AS THE CHANNEL SCHEDULER( IS INVOKED BY ISSUING THE STARTIO
*    MACRO.  PASSED AS A PARAMETER IS THE SRB ADDRESS.  THE SRB CON-
*    TAINS A POINTER TO THE IOSB.  THE TCB = PARAMETER IDENTIFIES THE
*    TCB UNDER WHICH THE I/O IS TO BE PERFORMED.  IF TCB=0 IS SPECI-
*    FIED, THE CURRENT TCB (PSATOLD) IS USED.  IF TCB-SRB IS SPECIFIED,
*    THE SRB ALREADY CONTAIN THE TCB ADDRESS.  THE SOLE PURPOSE OF THE
*    TCB= PARAMETERIS TO TERMINATE THAT TASK IF AN IOS ERROR OCCURS.
*    THE CHANNEL SCHEDULER INTERFACE GUARANTEES THAT REGISTER 13 WILL
*    NOT BE CHANGED, BUT ALL OTHER REGISTERS MAY BE.  THEREFORE ALL
*    REGISTERS WILL HAVE TO BE SAVED BY THE IOS INVOKER.  IF A USER-
*    PROVIDED SAVE AREA IS NOT AVAILABLE, THE ASXBFLSA SAVE AREA MAY
*    BE USED AS THE LOCAL LOCK IS HELD.
*
*
         SPACE 2
         XC    ECB,ECB                 CLEAR ECB
         STM   14,12,12(13)            CHANNEL SCHEDULER DESTROYS ALL
*                                      REGISTERS EXCEPT 13
         LR    1,4
         STARTIO SRB=(1),TCB=SRB
         LM    14,12,12(13)
         SPACE 2

Erratum: There should be a SETLOCK RELEASE,TYPE=LOCAL here 11/20/2008 PHH

*
*
*    WAIT FOR I/O COMPLETION.  THE ECB WAITED UPON IS POSTED IN THE
*    IOS POST STATUS TERMINATION ROUTINE.
*
*
         SPACE 2
         WAIT  ECB=ECB



         SPACE 2
*
*
*    UNFIX CHANNEL PROGRAM AND I/O AREA.  THE PGFREE MACRO IS USE.
*    IN HIGH PERFORMANCE APPLICATIONS THE PGFREE BRANCH ENTRY IS TO
*    BE RECOMMENDED.
*
*
         SPACE 2
         LA    1,FIX                   STARTING LOCATION
         LA    15,FIXEND               ENDING LOCATION + 1
         PGFREE R,A=(1),EA=(15)
         SPACE 2
*
*
*    PROGRAM CLEANUP.  THE SRB/IOSB IS FREED AND A DUMP IS TAKEN.  AS
*    THE DYNAMICALLY ACQUIRED AREA HAS NOT BEEN FREED, THE CHANNEL
*    PROGRAM, I/O AREA, ECB, AND SAVE AREA MAY BE EXAMINED BY THE
*    STUDENT.
*
         SPACE 2
*    FREE SRB/IOSB
         LR    1,4
         FREEMAIN RU,LV=SRBSIZE+IOSBSIZE,A=(1),SP=245
*    OBTAIN A DUMP
         CLI   ECB,X'7F'               Q/GOOD I/O
         BE    GOODIO
         ABEND 100,DUMP
GOODIO   DS    0H
         ABEND 200,DUMP
         SPACE 2
*
*
*      IOS POST STATUS NORMAL EXIT.  THIS ROUTINE IS A NO-OP.
*
*
         SPACE 2
STIONRMR EQU   *
         BR    14
         SPACE 2
*
*
*     IOS POST STATUS ABNORMAL EXIT.  THIS ROUTINE IS A NO-OP
*
*
         SPACE 2



STIOABNR EQU   *
         BR    14
         SPACE 2
*
*
*     IOS POST STATUS TERMINATION ROUTINE.  THIS ROUTINE IS EXECUTED
*     UNDER THE SRB PASSES AS A PARAMETER TO IOS.  THE LOCAL LOCK IS
*     HELD.  AT LEAST TWO THINGS MUST BE DONE IN A TERMINATION ROU-
*     TINE.  THE COMPLETION ECB MUST BE POSTED (USING THE POST BRANCH
*     ENTRY) AND THE PURGE ROUTINE MUST BE CALLED IF AN IPIB EXISTS.
*
*
         SPACE 2
STIOPSTR EQU   *
*   THE IOSOPT FLAGS SPECIFIES LOCAL LOCK HELD OPTION.  IT NEED NOT BE
*   OBTAINED HERE.  THE SETLOCK MACRO IS INCLUDED FOR ILLUSTRATIVE
*   PURPOSES ONLY.
*         SETLOCK OBTAIN,TYPE=LOCAL,   OBTAIN LOCAL LOCK
*               MODE=UNCOND,
*               REG=USE,
*               RELATED=(POST,RELEASE)
*        POST THE COMPLETION ECB
         L     3,FLCCVT                CVT ADDR.
         LR    5,1                     IOSB ADDR.
         USING IOSB,5
         BALR  6,0
         USING *,6
         L     13,IOSUSE               CORE ADDR.
         IC    10,IOSCOD               COMPLETION CODE
         SLL   10,24
         LA    11,ECB                  ECB ADDR.
         L     15,CVT0PT01-CVT(3)      POST BRANCH ENTRY ADDR.
         BALR  14,15                   POST ECB
*     INTERFACE WITH PURGE TO UPDATE QUIESCE COUNT.
         L     1,IOSIPIB
         LTR   1,1                     Q/IPIB PRESENT
         BZ    RELEASE                 NO
         SR    0,0                     INDICATE LOCAL LOCK HELD
         L     15,CVTIXAVL-CVT(3)
         L     15,IOCQCNT-IOCOM(15)
         BALR  14,15                   UPDATE QUIESCE COUNT
RELEASE  DS    0H
         SETLOCK RELEASE,TYPE=LOCAL,RELATED=(POST,OBTAIN)
         SPACE 2
*
*



*    RETURN TO DISPATCHER.  THE DISPATCHER PASSES IN REGISTER 14 THE
*    ADDRESS OF IT'S SRB RETURN ROUTINE IEAPDSRT.  THIS ROUTINE MAY
*    BE LOCATED VIA V=CVTSRBRT.
*
*
         SPACE 2
         L     14,CVTSRBRT-CVT(3)
         BR    14                      RETURN TO DISPATCHER
*     LITERALS
         LTORG
*  DSECTS
CVT      DSECT
         CVT   DSECT=YES
         IHAPSA
         IHAASCB
         IHASRB
         IECDIOSB
         IECDIOCM
IOSBSIZE EQU   IOSEND-IOSB
CORE     DSECT
         DS    18F
ECB      DS    F
FIX      EQU   *
SEARCH   DS    D
TIC      DS    D
READ     DS    D
LABEL    DS    CL80
FIXEND   EQU   *
CORESIZE EQU   *-CORE
         END   ,
//LKED.SYSLMOD  DD  DSN=xxxx.xxxxxxx(STARTIO),DISP=SHR,SPACE=
//LKED.SYSPRINT DD  SYSOUT=*


